

「EQUATRAN-M」技術計算用連立方程式解法言語

(9・最終回)

EQUATRAN-M 便利帳

横山 克己*

小口 梧郎**

宮原 昶中***

前回まで8回にわたり、EQUATRAN-Mについて各分野ごとに例題を取り上げて解説を進めてきた。今回は最終回ということで、全体のまとめと補足、そして知っておくと便利な事柄を述べて、EQUATRAN-Mを活用する上での参考に供する。

1. EQUATRAN-M の構成

本連載は、例題を中心としてEQUATRAN-Mの各機能を述べてきたので、読者の中には使い方、用語等で混乱している方もおられると思う。そこで、改めてEQUATRAN-M全体を振り返っておこう。

EQUATRAN-Mの構成は、図9-1のようになっている。解くべき方程式を入力したものがソーステキストであり、EQUATRAN-M内部で解析されて、オブジェクトと呼ぶ解法の手順を表わすテーブルに変換される(コンパイル)。このテーブルをもとに、必要なら入力データを取り込んで、数値計算を行って答えを出す(ゴー)。データの入力には二通りある。あらかじめソーステキストのようにテキスト(データテキストという)として用意した一括入力データと、実行時に対話的に入力するデータである。図に示したように、

ソーステキストやデータテキストを作成したり、修正したりする機能がエディタである。これらのテキストは、ディスクに保存したり、プリンタに出力したりできる。一方、得られた計算結果ももちろんプリントアウトでき、さらにディスクに出力できる(ディスクへの出力の使い方は後ほど説明する)。また、破線で示したように、同じ方程式を用い、一部の入力値を変えて次々に計算することをケーススタディと呼ぶ。実際の操作法やコマンドについては、この連載の第1回目の4節に




図 9-1 EQUATRAN-M の構成

* Katsumi Yokoyama 三井東圧化学(株)システム部
** Goro Oguchi 三井東圧化学(株)システム部 主査
*** Koreatsu Miyahara 三井東圧化学(株)システム部 次長

述べられているので、照らし合わせて読まれるとよいだろう。

2. 文法のまとめと補足

方程式を記述するときの決まりを文法と呼んでいるが、これは第1回と第2回で簡単に解説した。それら文法を使った特徴的な例題が、毎回登場したので、表9-1にまとめておく。なお、説明の項は回、節を表わし（たとえば2(3)ならば第2回3節）、例題はその番号を表わしている。

次に、論理式を用いた補足の例題として、攪拌所要動力を取り上げる。

<例題 9.1：攪拌所要動力>¹⁾

内径 $D=2.4\text{m}$ の槽内に比重 $\rho=1.2\text{g/cm}^3$ 、粘度 $\mu=200\text{cp}$ の液を内径の7割に等しい深さまで入れ、これに直径 $d=1.2\text{m}$ 、幅 $b=0.48\text{m}$ の2枚羽根のかい型攪拌機を挿入し、 $n=60\text{rpm}$ で攪拌した場合の動力を完全邪魔板条件と邪魔板なしの場合とについて求めよ。ただし、翼角度は $\theta=90^\circ, 45^\circ, 30^\circ$ の3種とする。

解) 次の永田らの式を用いて記述したソーステキストが、表9-2である。

表 9-1 文法一覧

項目	文	説明	例題
方程式の表記		(1)2注, (1)3.3	各例題
変数名		(1)3.3	
注釈(コメント)	/* */	(1)4.4	
入力	INPUT 文		
出力	OUTPUT 文		
組み込み関数		(1)3.1	1.1 2.1 3.6 4.7 8.1
配列変数の定義	VARIABLE 文 (省略形 VAR)	(2)4	2.4
配列変数の表記		(2)4	2.4 4.7 6.1 8.3
配列定数の表記		(2)4	2.4 6.1 7.6 7.8
配列変数間の演算		(2)4	2.4 4.7 6.1 7.8 8.3
繰り返し収束計算	RESET 文	(2)1	2.1 3.5 4.5 5.3 6.2
論理演算, 論理式		(2)2	2.2 4.6 4.8 8.1 9.1
条件付の式	WHEN	(2)2	2.2 4.8 8.1 8.2
数表	TABLE 文	(2)3	2.3 5.3 7.1 7.2 8.2
最適化計算	FIND 文	(2)6	2.6 4.6 5.2 7.5 7.7
パラメータ	GLOBAL 文	(2)5	2.5 4.4 6.1 6.5 9.2
	LOCAL 文	(2)5	7.1 9.2
マクロの定義	MACRO 文	(2)5	2.4 3.7 4.4 6.5 9.2
	END 文	(2)5	
マクロコール	CALL 文	(2)5	

表 9-2 例題9.1のリスト

```

1: /* かくはん所要動力 */
2: /* 化学工学便覧 改訂三版 p.1082 (1988)
3:
4: P_ : 所要動力 [PS]
5: Np : 動力数 [-]
6: Re : レイノルズ数 [-]
7: rho_ : 比重 [g/cm3]
8: mu_ : 粘度 [cp]
9: n_ : 回転数 [rpm]
10: D : 槽径 [m]
11: H : 液深さ [m]
12: d : 翼径 [m]
13: b : 翼幅 [m]
14: theta_ : 羽根角度 [°] */
15:
16: Np = P*gc/(rho*n^3*d^5)
17: Np = A/Re+B*((1e3+1.2*Re^0.66)/(1e3+3.2*Re^0.66))^p..
18:      *(H/D)^(0.35+b/D)*(sin(theta))^1.2
19: A = 14+(b/D)*(670*(d/D-0.6)^2+185)
20: B = 10^(1.3-4*(b/D-0.5)^2-1.14*(d/D))
21: p = 1.1+4*(b/D)-2.5*(d/D-0.5)^2-7*(b/D)^4
22:
23: Rc = (25/(b/D)*(d/D-0.4)^2+b/D/(0.11*(b/D)-0.0048))
24: Re = 10^(4*(1-sin(theta))*Rc*(jamaita != 0) ..
25:      *(d^2*n*rho/mu) *(jamaita == 0)
26:
27: gc = 9.80665 /* 重力換算係数 */
28:
29: P_ = P/75 /* [PS] -> [Kg*m/sec] */
30: n_ = n/60 /* [rpm] -> [1/sec] */
31: rho_ = rho/1000 /* [g/cm3] -> [kg/m3] */
32: mu_ = mu/1000 /* [cp] -> [kg*m/sec] */
33: theta_ = 3.141593/180 = theta /* [°] -> [-] */
34:
35: INPUT rho_, mu_, D, H, d, b, theta_, n_, jamaita
36: OUTPUT jamaita, theta_, P_

```

$$N_p = \frac{P_{gc}}{\rho n^3 d^5} = \frac{A}{Re} + B \left(\frac{10^3 + 1.2 Re^{0.66}}{10^3 + 3.2 Re^{0.66}} \right)^p \left(\frac{H}{D} \right)^{(0.35 + b/D)} (\sin \theta)^{1.2} \quad (9-1)$$

$$A = 14 + (b/D) \{ 670(d/D - 0.6)^2 + 185 \} \quad (9-2)$$

$$B = 10 \{ 1.3 - 4(b/D - 0.5)^2 - 1.14(d/D) \} \quad (9-3)$$

$$p = 1.1 + 4(b/D) - 2.5(d/D - 0.5)^2 - 7(b/D)^4 \quad (9-4)$$

完全邪魔板条件

$$Re = 10^4 (1 - \sin \theta) R_c \quad (9-5)$$

$$R_c = \frac{25}{(b/D)} \left(\frac{d}{D} - 0.4 \right)^2 + \left\{ \frac{b/D}{0.11(b/D) - 0.0048} \right\} \quad (9-6)$$

邪魔板なし

$$Re = d^2 n \rho / \mu \quad (9-7)$$

24~25行目に論理式を用い、入力変数 jamaita の値 (1 または 0) によって、完全邪魔板条件と邪魔板なしの場合に計算する式を振り分けている。すなわち、論理値の真は 1、偽は 0 と扱われるので、jamaita の値が 1 のときは (jamaita != 0) の値は 1、(jamaita == 0) の値は 0 となることを利用している。また、

表 9-3 例題 9.1 の入力データと計算結果

```

1: /* かくはん所要動力 データ */
2:
3: rho_ = 1.2          /* 比重 [g/cm3] */
4: mu_ = 200          /* 粘度 [cp] */
5: n_ = 60            /* 回転数 [rpm] */
6: D_ = 2.4          /* 槽径 [m] */
7: H_ = 1.68         /* 液深 [m] */
8: d_ = 1.2          /* 翼径 [m] */
9: b_ = 0.48         /* 翼幅 [m] */
10: jamaita = 1       /* 完全邪魔板条件 */
11: theta_ = 90       /* 羽根角度 [°] */
12: >go
13: theta_ = 45
14: >go
15: theta_ = 30
16: >go
17: jamaita = 0       /* 邪魔板なし */
18: theta_ = 90       /* 羽根角度 [°] */
19: >go
20: theta_ = 45
21: >go
22: theta_ = 30

```

```

[ 計算結果 ]
jamaita = 1
theta_ = 90
P_ = 24.166086

```

```

[ 計算結果 ]
jamaita = 1
theta_ = 45
P_ = 5.720129

```

```

[ 計算結果 ]
jamaita = 1
theta_ = 30
P_ = 2.582416

```

```

[ 計算結果 ]
jamaita = 0
theta_ = 90
P_ = 3.496797

```

```

[ 計算結果 ]
jamaita = 0
theta_ = 45
P_ = 2.315394

```

```

[ 計算結果 ]
jamaita = 0
theta_ = 30
P_ = 1.535958

```

29~33行では、式中で用いられる単位と入出力に使われる単位の換算を行っている。入出力に使われる単位での値を示す変数には、うしろに_を付けて区別した。

表 9-3 に、邪魔板の有無と翼角度についてのケーススタディを一度に行うためのデータテキストと計算結果を示す。2 ケース目以降については、変更する変数のデータのみを記述し、計算実行を指示する “>go” をはさんでケーススタディを一度に行っている。なお、記述されていない変数には前回の値が用いられる。

このほかに、論理式は答えが YES, NO となるような問題にも応用できる。

3. 応用分野のまとめと補足

今度は、数学的な分野でこの連載を見直してみよう。表 9-4 に取り上げられている項目を一覧にしてみた。やはり右側に例題の番号を示した。さまざまな分野が見られ、EQUATRAN-M の記述

性の広さがうかがえる。

ここでは、補足として Simpson 則による求積と、偏微分方程式の例題を一題ずつ紹介しよう。

<例題 9.2 : 断熱型回分反応器の設計>²⁾

A → R で表わされる液相反応を断熱型の回分反応器を用いて行う。反応速度は次式で表わされる。

$$-r_A = k_0 \exp(-E/RT) C_A^n$$

ここで $-r_A$ は反応速度 [g-mol/l·hr]、 k_0 は 1.5×10^{11} 、 E は 15,300 cal/g-mol、 R は気体定数で 1.987 cal/g-mol·K、 T は反応温度 [K]、 C_A は A の濃度 [g-mol/l]、 n は反応次数で 1.5 である。本反応を断熱的に行って、A の反応率が $x_A = 80\%$ に達するに必要な反応時間 θ [hr] を求めよ。ただし、初期反応温度 $T_0 = 293\text{K}$ 、A の初期濃度 $C_{A0} = 0.2 \text{ g-mol/l}$ 、反応液の比熱 $c_p = 0.95 \text{ cal/g·K}$ 、反応液密度 $\rho = 1.1 \text{ g/cm}^3$ 、反応熱 $-\Delta H_r = 35,000 \text{ cal/g-mol}$ とする。

解) 回分反応の基礎式によれば反応時間 θ は

$$\begin{aligned} \theta &= C_{A0} \int_0^{x_A} \frac{dx_A}{(-r_A)} \\ &= C_{A0} \int_0^{x_A} \frac{dx_A}{k_0 \exp(-E/RT) C_{A0}^n (1-x_A)^n} \end{aligned} \quad (9-8)$$

表 9-4 応用分野の一覧

項 目	例 題
線形連立方程式	1.3 ほか
非線形連立方程式	2.1 3.5 4.5 5.3 6.2 ほか
常微分方程式	4.7 8.3
偏微分方程式	9.3
一変数最適化問題	2.6 7.5
多変数最適化問題	1.4 4.6 5.2 7.7
求 積	9.2
2点境界値問題	8.3
固有値, 固有ベクトル	4.7 7.8
逆行列, 転置行列	4.7
平均, 分散	7.1
回帰分析	7.2
相関係数, 偏回帰係数	7.2
偏相関係数	7.3
重回帰分析	7.4 7.5
正規方程式	7.4 7.5
ラグランジュの未定乗数法	7.6
主成分分析	7.8
漸化式	8.3 9.2
複素数	8.1

表 9-5 例題 9.2 の解

```

1: /* 断熱型回分反応器の設計 */
2:
3: /* 化学工学プログラミング演習 p.88 培風館 (1976) */
4:
5: GLOBAL k0 = k0, E = E, rho = rho, cp = cp, ..
6: dH = dH, T0 = T0, Ca0 = Ca0, R = R
7: LOCAL n1 = 11, n2 = 10
8: VAR x1(n1), x2(n2), y1(n1), y2(n2)
9:
10: MACRO func
11:   VAR T(m)
12:   y = 1 / ( k0*exp(-E/R/T)+Ca0^0.5*(1-x)^1.5 )
13:   T = T0 + dH*Ca0/rho/cp*x
14: END func
15:
16: h = ( xend - 0 ) / (2*n2)
17: x1(1) = 0
18: x2(1:n2) = x1(1:n2) + h
19: x1(2:n1) = x2(1:n2) + h
20: f1: CALL func( x=x1, y=y1, m=n1 )
21: f2: CALL func( x=x2, y=y2, m=n2 )
22: theta = 1/3 * h *
23:   + ( y1(1)+4*sum(y2)+2*sum(y1(2:n1-1))+y1(n1) )
24:
25: k0 = 1.5e11 /* 反応速度定数 [-] */
26: E = 15300 /* 活性化エネルギー [cal/g-mol] */
27: rho = 1100 /* 反応液の密度 [g/l] */
28: cp = 0.95 /* 反応液の比熱 [cal/g-K] */
29: dH = 35000 /* 反応熱 [cal/g-mol] */
30: T0 = 293 /* 初期反応温度 [K] */
31: Ca0 = 0.2 /* 濃度 [g-mol/l] */
32: R = 1.987 /* 気体定数 [cal/g-mol-K] */
33:
34: xend = 0.8 /* 目的反応率 [-] */
35:
36: OUTPUT f1.T(n1) /* 最終温度 [K] */
37: OUTPUT theta /* 反応時間 [hr] */

```

[計算結果]
f1.T(11) = 298.358852
theta = 6.938563

また、断熱反応の場合、温度 T と反応率 x_A の間には

$$T = T_0 + \frac{(-\Delta H_r) C_{A0}}{\rho c_p} x_A \quad (9-9)$$

の関係がある。(9-9)式を(9-8)式に代入して x_A について積分をすることになる。数値積分の手法として Simpson 則を用いる。

$$S = \int_a^b f(x) dx \quad (9-10)$$

は

$$S = \frac{1}{3} h [f(a) + 4f(a+h) + 2f(a+2h) + 4f(a+3h) + 2f(a+4h) + \dots + 4f(b-h) + f(b)] \quad (9-11)$$

として計算できる。ここで $h=(b-a)/n$, n は分割数である。

これらをソーステキストにして答えを求めたのが表 9-5 である。被積分関数をマクロとして定義してあり (10~14行), 他の求積の問題もこのマクロを書き換えればよく応用がきく。ただし, マクロ内で使う変数 (定数として与えられるもの) は GLOBAL 文でパラメータ指定しておく (5~6行)。x1, y1 および x2, y2 はそれぞれ奇

表 9-6 例題 9.3 のリスト

```

1: /* 2次元熱伝導 (Laplace の方程式) */
2:
3: LOCAL n = 10, m = 10
4: VAR T(n,m)
5:
6: (T(3:n,2:m-1)-2*T(2:n-1,2:m-1)+T(1:n-2,2:m-1))/dx^2...
7: +(T(2:n-1,3:m)-2*T(2:n-1,2:m-1)+T(2:n-1,1:m-2))/dy^2...
8: = 0
9: dx = 1/(n-1)
10: dy = 1/(m-1)
11:
12: /* 境界条件 */
13: T(1:n,1) = 100
14: T(1:n,m) = 0
15: T(1,n,2:m-1) = 0
16:
17: OUTPUT T

```


表 9-7 例題 9.3 の計算結果

計算結果		(1)	(2)	(3)	(4)	(5)
1)	100	0	0	0	0	0
2)	100	48.629073	27.602587	17.231052	11.279234	20.408183
3)	100	66.913706	44.550224	30.042386	28.535057	28.535057
4)	100	74.475527	53.642216	37.980085	28.535057	28.535057
5)	100	77.346185	57.583028	41.700683	28.535057	28.535057
6)	100	77.346185	57.583028	41.700683	28.535057	28.535057
7)	100	74.475527	53.642216	37.980085	28.535057	28.535057
8)	100	66.913706	44.550224	30.042386	20.408183	20.408183
9)	100	48.629073	27.602587	17.231052	11.279234	11.279234
10)	100	0	0	0	0	0
		(6)	(7)	(8)	(9)	(10)
1)	0	0	0	0	0	0
2)	7.477702	4.855519	2.919055	1.370927	0	0
3)	13.778058	9.025316	5.449778	2.584651	0	0
4)	18.193023	12.014915	7.290082	3.437903	0	0
5)	20.441065	13.571237	8.252732	3.898878	0	0
6)	20.441065	13.571237	8.252732	3.898878	0	0
7)	18.193023	12.014915	7.290082	3.437903	0	0
8)	13.778058	9.025316	5.449778	2.584651	0	0
9)	7.477702	4.855519	2.919055	1.370927	0	0
10)	0	0	0	0	0	0

数番目および偶数番目の独立変数と関数値であり, それぞれについて上記のマクロを呼んでいる。また, 17~19行で x1, x2 に漸化的に値を与えている。21~22行の CALL 文の前に付けた f1, f2 はラベルである。最終温度は, このラベルを使って f1.T(n1) で表わされる (36行)。

<例題 9.3: 2次元熱伝導>

右図のような, 1辺だけ 100°C に接する境界をもったある物体内部での定常状態での温度分布を求めよ。ただし, 面に垂直方向の温度分布は無視できるものとする。



解) 2次元の熱伝導の基本式は次のような偏微分方程式で表わされる。

$$\frac{\partial^2 T}{\partial x^2} + \frac{\partial^2 T}{\partial y^2} = 0 \quad (9-12)$$

これは Laplace の方程式とも呼ばれ, 楕円型偏微分方程式の最も簡単なものである。この式を有限差分により置き換えると次のようになる。

$$\frac{T_{i+1,j} - 2T_{i,j} + T_{i-1,j}}{(\Delta x)^2} + \frac{T_{i,j+1} - 2T_{i,j} + T_{i,j-1}}{(\Delta y)^2} = 0 \quad (9-13)$$

$$(i=2, 3, \dots, n-1, j=2, 3, \dots, m-1)$$

表 9-8 診断メッセージ

```
[ 診断メッセージ ]
(1) 式と変数の数
    式 ..... 13 (E)
    全変数 ..... 22 (V)
    入力変数 ..... 9 (Vi)
    使われていない変数 ..... 0 (Vn)
(2) 定義されていない変数
    d          n          rho          gc
    p          np         Re           D
    b          theta      H            p
    B          A          Rc           ja_maita
    mu         p_         rho_
    mu         theta_
(5) 式と変数の数の一致
    (E)-(V)-(Vi)-(Vn) = 0 ... OK
(6) RESET 及び FIND
(7) 入力すべき変数
    rho_          mu_          D          H
    d             b            theta_     n_
    ja_maita
```

x 方向の分割数を n , y 方向の分割数を m とし, x 方向, y 方向の長さをともに単位長さ (=1) とするとソーステキストは表 9-6 と表記できる. その計算結果は表 9-7 のようになる.

4. エラー対処

EQUATRAN-M には, 方程式の記述に誤りをおかしたときや, 計算実行中にエラーが起きたときに手助けしてくれる機能がいくつか備わっているので以下に説明する. ①と②はコンパイルオプションで, ③と④はゴーオプションである. オプションについては第 2 回の 7 節参照.

①診断メッセージ

式と変数の不一致のエラーが出たら, オプションで指定し診断メッセージを出力すると, より多くの情報が得られる. 例として, 例題 9.1 の診断メッセージを表 9-8 に示した.

表 9-9 トレースと変数値の表示

```
1: /* 演算不能エラーの例 */
2:
3: 2 / ( x - 1 ) = y
4: x + 1 = a * 2
5: a = 1
```

トレースの出力

```
&4 a*2          = 2
&8 x+1         = 2
&16 x          = 1
&22 x-1        = 0
&28 2/(x-1)    = *3.59539E+308
&32 y          = *3.59539E+308
```

VALUEコマンドによる変数値の表示

```
変数名>y
y = *3.595386E+308
      ゼロでの割算 エラーが起きています
      x-1
```

②ソースリストー 2

このリストを出力することにより, マクロやパラメータが意図した通りに展開されているかどうかをチェックできる. すでに例題 6.5 で解説した.

③ダンプ

この機能は繰り返し収束計算や最適化計算の途中経過を出力して, 収束状況を確認するのに使う. 第 3 回の 4 節参照.

④トレース

計算実行時に, 逐次計算された変数の変数名と値とを表示するのがトレース機能である. ユーザーはこの機能を使って計算過程をなぞり, どこで望ましくない状態が発生しているかを突きとめることができる. 表 9-9 に次の VALUE コマンドと合せて例を示した.

⑤変数値の表示 (VALUE コマンド)

計算終了後に, 任意の変数の値を変数名を指定することによって表示することができる. この機能を用いると, 変数値とともに演算不能のエラーの原因を知ることができる. もし, 指定した変数が演算不能エラーに影響されている場合には, その変数値の前に *印が付けられるとともに, エラーの原因を示すメッセージとその原因となった変数名または演算式が表示される.

5. 他のプログラムとの関係

ここでは, BASIC プログラムとのデータのやり取りについて解説する. たとえば, 計測機器で得られたアナログデータをパソコンに取り込み, BASIC で加工したのち EQUATRAN-M で計算するという使い方や, EQUATRAN-M で計算した結果を出力し, BASIC で読み込んでグラフ表示などの使い方が考えられる. EQUATRAN-M にとって前者は入力であり, 後者は出力であるので分けて説明しよう. BASIC は MS-DOS 上で動くものが必要であるが, もしそうでないときはファイルコンバータが必要になる.

入力は, 表 9-3 に示したようなデータテキストの書式で, そのままアスキー形式のファイルとして作成すればよい. ただし, ファイル名の拡張子は EQD とし, 一行あたり 72 文字までとする. こ

表 9-10 ケーススタディの結果をグラフ化する BASIC プログラム

```

100 SCREEN 3,0:CLS 3:CONSOLE 0,25,0,1:WIDTH 80,25
110 DEFINT A-W
120 '
130 '   ファイル名   ケース数
140 READ FILES$, CASE
150 DATA "TEMP.EQS", 3
160 '
170 '   変数名     下限値   上限値   目盛り間隔
180 READ XS$, XL, XH, XW
190 READ Y$, YL, YH, YW
200 DATA "theta_", 0, 120, 10 'X軸
210 DATA "P_", 0, 30, 5 'Y軸
220 '
230 DIM X(CASE), Y(CASE)
240 OPEN FILES$ FOR INPUT AS #1
250 FOR I=1 TO CASE
260   ITEMS=XS$
270   GOSUB *READ.SCALAR : IF RET=1 THEN *ER
280   X(I)=Z
290   ITEMS=Y$
300   GOSUB *READ.SCALAR : IF RET=1 THEN *ER
310   Y(I)=Z
320 NEXT
330 CLOSE
340 GOSUB *GRAPH
350 LOCATE 0,23 : AS=INPUT$(1)
360 END
370 '-----
380 *ER
390 CLOSE
400 PRINT "Read Error"
410 END
420 '-----
430 *READ.SCALAR ' スカラー変数の値を読み取る
440 '           ' ITEMS に読み取る変数名を入れる
450 '           ' Z に値がセットされる
460 '           ' RET=1 の時は読み取りエラー
470 RET=0 : LENGTH=LEN(ITEMS) : ITEMS=ITEMS+"="
480 IF EOF(1)<>0 THEN RET=1 : RETURN
490 LINE INPUT #1, AS$
500 IF LEFT$(AS$,LENGTH+2)<>ITEMS THEN 480
510 Z=VAL(MID$(AS$,LENGTH+5))
520 RETURN
530 '-----
540 *GRAPH ' グラフを表示する
550 '       ' X(I) にX軸のデータを入れる
560 '       ' Y(I) にY軸のデータを入れる
570 DEF FN(X)=56+(X-XL)/(XH-XL)*568
580 DEF FN(Y)=375-(Y-YL)/(YH-YL)*367
590 LINE(FN(XL),FN(YL))-(FN(XH),FN(YH)),5,B
600 FOR XI=XL TO XH STEP XW
610   LINE(FN(XI),FN(YL))-(FN(XI),FN(YL)-5),5
620   LOCATE FN(XI)/8-2,24:PRINT USING "###":XI:
630 NEXT
640 FOR YI=YL TO YH STEP YW
650   LINE(FN(XL),FN(YI))-(FN(XL)+5,FN(YI)),5
660   LOCATE 2,FN(YI)/16-.4:PRINT USING "###":YI
670 NEXT
680 FOR I=1 TO CASE
690   CIRCLE(FN(X(I)),FN(Y(I))),3,7
700   PAINT (FN(X(I)),FN(Y(I))),7,7
710   IF I>1 THEN LINE(FN(X(I-1)),FN(Y(I-1)))
       -(FN(X(I)),FN(Y(I))),
       7.,&H8888
720 NEXT
730 COLOR 8:LOCATE 70,22:PRINT XS$:
740 LOCATE 8,1:PRINT Y$::COLOR 7
750 RETURN

```

のファイルを EQUATRAN-M の中でデータテキストとしてロードしてやればよい。

出力は、EQUATRAN-M の計算結果の出力先をオプションでディスクに指定してやれば、一時ファイルという、プリンタに出力されるものと同じイメージのアスキー形式のファイルが作られるので、これを利用することができる。BASIC でこのファイルを読み取るには少し工夫がいるが、たとえば表9-10のようにすればよい(このサンプルは PC9801 シリーズの N₈₈-BASIC を使用した)。このプログラムは、例題9.1のケーススタディ

の結果をグラフ化するものである。430~520行のサブルーチンでスカラー変数を読み取っている。

6. EQUATRAN-M では扱いにくい問題

EQUATRAN-M は汎用性の高い方程式解法言語であるが、もちろん万能ではない。EQUATRAN-M を有効に使うためにはその限界をよく知っておくことも大切である。

常微分方程式、偏微分方程式あるいは積分方程式は直接表現することができないので差分近似をすることが必要となる。差分近似も区間の分割を細かくしすぎると変数の数が多くなりパソコンで解ける範囲を越えてしまう(メモリが不足してしまう)。常微分方程式の例題8.3はややその限界に近い。偏微分方程式については、例題9.3で取り上げたが実用的な規模の問題を扱うのはちょっと無理であろう。

複素数や整数を変数とする方程式も直接扱うことはできない。複素数については、例題8.1のように実部と虚部とを別々の変数として取扱えばある程度対応が可能であるが、EQUATRAN-M の特徴である記述性の良さは失われてしまうことが多い。

数値計算上特殊な手法を必要とする問題も EQUATRAN-M の不得手とするところである。ある種の非線形方程式はニュートンラフソン法(EQUATRAN-M の収束計算手法)では非常に解きにくいことが知られている。また最適化については、線形計画問題も EQUATRAN-M 向きでないといえる。大規模な問題あるいは特殊なアルゴリズムによる必要のあるものについては他の解説を参照されたい³⁾。

EQUATRAN-M は方程式を解くことはできるが方程式を作ることにはできない。したがって方程式の導出そのものに手続き的なプログラムを必要とする問題も不向きである。有限要素法などもこの種の問題にあたる。

7. おわりに

EQUATRAN-M は方程式解法のための言語であり、また大変使いやすい言語である。その使い

表 9-11 EQUATRAN-M 機能一覧

機 能	内 容
I. 方程式記述機能	通常の数学的記述法による (等号は左辺と右辺が等しい意味で使用される) 算術演算子 5種 (+, -, *, /, ^) 比較演算子 6種 (>, <, >=, <=, =, !=) 論理演算子 3種 (!, &,) 初等関数など 24種 変数の値によって式の形が異なる式を記述できる 一次元 (ベクトル) および二次元 (マトリックス) の配列変数、配列定数が可能 配列要素間の演算、配列と配列との演算が可能 次元の異なる配列間、配列とスカラー変数間の演算が可能 配列への組み込み関数の適用が可能 添字の組み合わせにより部分配列が記述できる 一次元 (引数が1つ) および二次元 (引数が2つ) の数表を定義可能 直線内挿、外挿による補間 数表逆引き機能 一次元および多変数の最大および最小探索問題を記述可能 陰な不等式制約条件が可能 任意の方程式群をマクロとして登録し、これをマクロコールにより随時呼び出し可能 (マクロコールのネスタングは最大5レベルまで) 任意の文字列をパラメータにより置換
II. 方程式解法機能	最適計算手順 (計算量最少) の自動生成と実行 線形連立方程式の自動判別による解法の組み込み 非線形連立方程式の自動判別と収束計算の自動組み込み ユーザーによる収束方法の指定機能 ガウスの消去法 (パーシャルピボットリング) ニュートンラフソン法 (数値微分による) 一次数...二次曲線当てはめ法 多変数...Bogx のコンプレックス法 倍精度 (10進で15桁、10 ⁻³⁰⁷ ~10 ⁺³⁰⁸) サポートあり 最大4000変数 (中間変数を含むソフトウェア上の制限、使用可能なメモリ量によっても制約される)
III. 実行制御	システムコマンド20種 ソーステキスト、データテキストのロード、セーブ オブジェクトファイルのロード、セーブ データテキストによる一括ケーススタディ 対話入力によるケーススタディ
IV. エディタ	フルスクリーンテキストエディタを内蔵 エディットコマンド8種 ソーステキスト (方程式) 最大600行 データテキスト (ケーススタディデータ) 最大300行
V. その他の機能	計算結果をMS-DOSのアスキーファイルへ出力 (他のプログラムへの入力可能) ヘルプ機能 コマンドの使用法、方程式記述文法、特殊キーの使用法などを表示 診断メッセージ、収束過程の表示、実行時トレース 例外演算 (ゼロ割りなど) の原因と原因変数名の表示

表 9-12 EQUATRAN-M 発表文献一覧

1. アプリケーション・プログラム—化工計算: bit 9 No. 9, 987 (1977)
2. Computer programs for chemical engineers: *Chem. Eng'g, Aug. 28* 109 (1978)
3. G. Oguchi, M. Mitsunaga: A powerful language to solve a set of nonlinear equations, International congress "Contribution of computers to the development of chemical engineering and industrial chemistry", at Paris (March 1978)
(以上は大型計算機用 EQUATRAN)
4. 小口, 横山, 佐藤: プロセス 物質収支計算への方程式解法言語の応用, プロセスシステム工学総合シンポジウム, 1985.3, 於芦屋
5. 小口, 横山, 佐藤: パソコン用方程式解法言語の応用, 化学工学協会第50年会, E302 1985.3, 於横浜
6. 宮原, 林田, 須藤: 分離における数値計算, 分離技術, No.5, p.281 (1985)
7. 宮原, 林田, 須藤: 分離技術懇話会夏季研究討論会資料, 1985.8 於浜松
8. 小口: 創造的技術者の生産性を向上させる方程式解法ソフト—EQUATRAN-M, 日経コンピュータ 9月30日号 p.207 (1985)
9. 宮原, 小口: 方程式解法ソフト EQUATRAN-M, OR学会秋期研究発表会, アブストラクト集 (SW-9), p.228 (1985)
10. 宮原他: 連載「EQUATRAN-M」技術計算用連立方程式解法言語, ケミカル・エンジニアリング 8月号 (1985) ~4月号 (1986)

やすさの秘密は2つある。1つは記述性の良さである。例えば BASIC で書けば50行を要する問題を EQUATRAN-M では10行足らずで書くことができる。2つめは EQUATRAN-M が宣言型の言語であることである。すなわち EQUATRAN-M で書かれた文 (方程式) は一つ一つが独立して変数間に一つの関数関係が成り立つべきことを宣言している。この関係が、他の手続型のプログラミング言語の場合のように、別の記述によって変更されたりあるいは無効にされてしまうようなことは起らない。したがって計算結果は (数値計算上のエラーが生じない限り) 常に全ての方程式を満たしている。このことは方程式の記述における誤りの防止と発見を容易にすると共に、計算結果の信頼性を著しく高めている。

一方, EQUATRAN-Mにはまだまだ発展の余地も大きい。前項に掲げた問題, 特に常微分方程式の取扱いや, あるいは計算結果のグラフ化機能などはユーザーからの要望も多い。これらの機能強化も順次行っていきたいと考えている。

最後に EQUATRAN-M の機能一覧表 (表9-

11) と EQUATRAN-M に関する文献の一覧 (表9-12) を掲げて本連載を終了する。

参考文献

- 1) 化学工学協会編: 化学工学便覧, 改訂三版, p.1082 丸善 (1968).
- 2) 化学工学協会編: 化学工学プログラミング演習, p.68, 培風館 (1976).
- 3) 宮原: コンピュートロール, No. 12, p.49 (1985).